ME 406
Assignment #8 Solutions

■ Problem 1

Let the linearly independent generalized eigenvectors of **A** associated with λ be $\mathbf{V}^{(i)}$, $i = 1, m$. Let the solution of the equations formed from $\mathbf{V}^{(i)}$be $\mathbf{X}^{(i)}$. We wish to show that the $\mathbf{X}^{(i)}$ form a linearly independent set. Suppose otherwise. Then there exist constants a_i not all zero such that

$$a_1\mathbf{X}^{(1)} + a_2\mathbf{X}^{(2)} + \ldots + a_m\mathbf{X}^{(m)} = 0 \text{ for all } t.$$

Let $t = 0$. Then the above reduces to

$$a_1\mathbf{V}^{(1)} + a_2\mathbf{V}^{(2)} + \ldots + a_m\mathbf{V}^{(m)} = 0 \ .$$

Because the set of $\mathbf{V}^{(i)}$is linearly independent, it follows that all a_i= 0. Therefore the solutions $\mathbf{X}^{(i)}$are linearly independent.

■ Problem 2

We enter the matrix **A** and the vector **b** into *Mathematica*.

```
A = {{4, 1, 2}, {0, 4, 0}, {2, 1, 4}};
```

```
b = {3, -1, 2};
```

We find a particular solution first.

```
Xp = -Inverse[A].b
```

$$\left\{-\frac{17}{24}, \frac{1}{4}, -\frac{5}{24}\right\}$$

(a) Now we solve the initial value problem. To solve the homogeneous equation, we find the eigenvalues and eigenvectors of **A**.

```
eigans = Eigensystem[A]
```

```
{{6, 4, 2}, {{1, 0, 1}, {1, -2, 1}, {-1, 0, 1}}}
```

The eigenvalues are distinct so this will be straightforward. We name the eigenvectors and eigenvalues.

```
λ1 = First[First[eigans]]
```

```
6
```

```
λ2 = First[eigans][[2]]
```

4

```
λ3 = First[eigans][[3]]
```

2

```
v1 = Last[eigans][[1]]
```

{1, 0, 1}

```
v2 = Last[eigans][[2]]
```

{1, -2, 1}

```
v3 = Last[eigans][[3]]
```

{-1, 0, 1}

The solution of the homogeneous equation is then

```
Xh = a1 * Exp[λ1 * t] * v1 + a2 * Exp[λ2 * t] * v2 + a3 * Exp[λ3 * t] * v3
```

$$\left\{-a3\, e^{2t} + a2\, e^{4t} + a1\, e^{6t},\ -2\, a2\, e^{4t},\ a3\, e^{2t} + a2\, e^{4t} + a1\, e^{6t}\right\}$$

The general solution is the sum of the homogeneous and particular solutions.

```
Xgen = Xp + Xh
```

$$\left\{-\frac{17}{24} - a3\, e^{2t} + a2\, e^{4t} + a1\, e^{6t},\ \frac{1}{4} - 2\, a2\, e^{4t},\ -\frac{5}{24} + a3\, e^{2t} + a2\, e^{4t} + a1\, e^{6t}\right\}$$

To solve the initial value problem, we must find values of a1, a2, and a3 such that Xgen at $t = 0$ takes on the value

```
Xinit = {0, 0, 0};
coeff = Solve[(Xgen /. t → 0) == Xinit, {a1, a2, a3}]
```

$$\left\{\left\{a1 \to \frac{1}{3},\ a2 \to \frac{1}{8},\ a3 \to -\frac{1}{4}\right\}\right\}$$

Now we put these values back into Xgen to get our solution. We do it in such a way as to preserve the expression for Xgen, so that we can use it in the next solution.

```
Xsol1 = Xgen /. Thread[Flatten[coeff]]
```

$$\left\{-\frac{17}{24}+\frac{e^{2t}}{4}+\frac{e^{4t}}{8}+\frac{e^{6t}}{3},\ \frac{1}{4}-\frac{e^{4t}}{4},\ -\frac{5}{24}-\frac{e^{2t}}{4}+\frac{e^{4t}}{8}+\frac{e^{6t}}{3}\right\}$$

We check our result, both for the equation and the initial condition.

```
D[Xsol1, t] - A.Xsol1 - b
```

$$\left\{-\frac{13}{4}+\frac{e^{2t}}{2}+\frac{3e^{4t}}{4}+2e^{6t}-2\left(-\frac{5}{24}-\frac{e^{2t}}{4}+\frac{e^{4t}}{8}+\frac{e^{6t}}{3}\right)-4\left(-\frac{17}{24}+\frac{e^{2t}}{4}+\frac{e^{4t}}{8}+\frac{e^{6t}}{3}\right),\ 1-e^{4t}-4\left(\frac{1}{4}-\frac{e^{4t}}{4}\right),\ -\frac{9}{4}-\frac{e^{2t}}{2}+\frac{3e^{4t}}{4}+2e^{6t}-4\left(-\frac{5}{24}-\frac{e^{2t}}{4}+\frac{e^{4t}}{8}+\frac{e^{6t}}{3}\right)-2\left(-\frac{17}{24}+\frac{e^{2t}}{4}+\frac{e^{4t}}{8}+\frac{e^{6t}}{3}\right)\right\}$$

```
Simplify[%]
```

{0, 0, 0}

```
Xsol1 /. t → 0
```

{0, 0, 0}

Everything checks.

(b) This is a repeat with a different initial condition. If you are observant you will know to expect a particularly simple solution. However, we can also let our machinery tell us that.

```
Xinit = {-17 / 24, 1 / 4, -5 / 24};

Xsol2 =
 Xgen /. Thread[Flatten[Solve[(Xgen /. t → 0) == Xinit, {a1, a2, a3}]]]
```

$$\left\{-\frac{17}{24},\ \frac{1}{4},\ -\frac{5}{24}\right\}$$

This answer is predictable if you happen to notice that the initial condition is equal to the time-independent particular solution.

■ Problem 3

We enter the matrix A and ask *Mathematica* for an eigenanalysis.

```
A = {{-1, 4, -1}, {-2, -2, 2}, {1, -4, -3}};
```

```
{λ1, λ2, λ3} = Eigenvalues[A]
```

$\{-2+4\,i,\ -2-4\,i,\ -2\}$

```
{v1, v2, v3} = Eigenvectors[A]
```

$\left\{\left\{-\frac{3}{5}+\frac{4\,i}{5},\ -\frac{2}{5}-\frac{4\,i}{5},\ 1\right\},\ \left\{-\frac{3}{5}-\frac{4\,i}{5},\ -\frac{2}{5}+\frac{4\,i}{5},\ 1\right\},\ \{1,\ 0,\ 1\}\right\}$

We have one real eigenvalue and two complex conjugate eigenvalues. We form three linearly independent real-valued solutions as a solution basis.

```
X1 = Exp[λ3 * t] * v3
```

$\left\{e^{-2t},\ 0,\ e^{-2t}\right\}$

```
X2 = Simplify[Re[ComplexExpand[Exp[λ2 * t] * v2]], t ∈ Reals]
```

$\left\{-\frac{1}{5}e^{-2t}(3\,\text{Cos}[4t]+4\,\text{Sin}[4t]),\ -\frac{2}{5}e^{-2t}(\text{Cos}[4t]-2\,\text{Sin}[4t]),\ e^{-2t}\,\text{Cos}[4t]\right\}$

```
X3 = Simplify[Im[ComplexExpand[Exp[λ2 * t] * v2]], t ∈ Reals]
```

$\left\{\frac{1}{5}e^{-2t}(-4\,\text{Cos}[4t]+3\,\text{Sin}[4t]),\ \frac{2}{5}e^{-2t}(2\,\text{Cos}[4t]+\text{Sin}[4t]),\ -e^{-2t}\,\text{Sin}[4t]\right\}$

Now we construct the solution satisfying the given initial conditions.

```
Xinit = {1, 3, 1};
Xgen = a1 * X1 + a2 * X2 + a3 * X3
```

$\left\{a1\,e^{-2t}+\frac{1}{5}a3\,e^{-2t}(-4\,\text{Cos}[4t]+3\,\text{Sin}[4t])-\frac{1}{5}a2\,e^{-2t}(3\,\text{Cos}[4t]+4\,\text{Sin}[4t]),\ -\frac{2}{5}a2\,e^{-2t}(\text{Cos}[4t]-2\,\text{Sin}[4t])+\frac{2}{5}a3\,e^{-2t}(2\,\text{Cos}[4t]+\text{Sin}[4t]),\ a1\,e^{-2t}+a2\,e^{-2t}\,\text{Cos}[4t]-a3\,e^{-2t}\,\text{Sin}[4t]\right\}$

```
Xsol2 = Simplify[
  Xgen /. Thread[Flatten[Solve[(Xgen /. t → 0) == Xinit, {a1, a2, a3}]]]]
```

$$\left\{\frac{1}{2} e^{-2t} (5 - 3\,\mathrm{Cos}[4\,t] + 6\,\mathrm{Sin}[4\,t]),\ 3 e^{-2t}\,\mathrm{Cos}[4\,t],\ -\frac{1}{2} e^{-2t} (-5 + 3\,\mathrm{Cos}[4\,t] + 6\,\mathrm{Sin}[4\,t])\right\}$$

We check our result.

```
D[Xsol2, t] - A.Xsol2
```

$$\left\{-12 e^{-2t}\,\mathrm{Cos}[4\,t] - \frac{1}{2} e^{-2t} (5 - 3\,\mathrm{Cos}[4\,t] + 6\,\mathrm{Sin}[4\,t]) - \frac{1}{2} e^{-2t} (-5 + 3\,\mathrm{Cos}[4\,t] + 6\,\mathrm{Sin}[4\,t]) + \frac{1}{2} e^{-2t} (24\,\mathrm{Cos}[4\,t] + 12\,\mathrm{Sin}[4\,t]),\right.$$
$$-12 e^{-2t}\,\mathrm{Sin}[4\,t] + e^{-2t} (5 - 3\,\mathrm{Cos}[4\,t] + 6\,\mathrm{Sin}[4\,t]) + e^{-2t} (-5 + 3\,\mathrm{Cos}[4\,t] + 6\,\mathrm{Sin}[4\,t]),$$
$$\left.12 e^{-2t}\,\mathrm{Cos}[4\,t] - \frac{1}{2} e^{-2t} (24\,\mathrm{Cos}[4\,t] - 12\,\mathrm{Sin}[4\,t]) - \frac{1}{2} e^{-2t} (5 - 3\,\mathrm{Cos}[4\,t] + 6\,\mathrm{Sin}[4\,t]) - \frac{1}{2} e^{-2t} (-5 + 3\,\mathrm{Cos}[4\,t] + 6\,\mathrm{Sin}[4\,t])\right\}$$

```
Simplify[%]
```

{0, 0, 0}

```
(Xsol2 /. t → 0) - Xinit
```

{0, 0, 0}

■ Problem 4

We enter the matrix and ask *Mathematica* for the eigenanalysis.

```
A = {{0, 1 / 4, 3 / 2, 3 / 4, -7 / 4},
   {3 / 2, 3 / 4, -1 / 2, -5 / 4, 11 / 4}, {1, -1 / 4, -1 / 2, -3 / 4, 7 / 4},
   {5, 19 / 4, -1 / 2, -15 / 4, 19 / 4}, {5 / 2, 5 / 2, 1, -2, 5 / 2}};
```

```
MatrixForm[A]
```

$$\begin{pmatrix} 0 & \frac{1}{4} & \frac{3}{2} & \frac{3}{4} & -\frac{7}{4} \\ \frac{3}{2} & \frac{3}{4} & -\frac{1}{2} & -\frac{5}{4} & \frac{11}{4} \\ 1 & -\frac{1}{4} & -\frac{1}{2} & -\frac{3}{4} & \frac{7}{4} \\ 5 & \frac{19}{4} & -\frac{1}{2} & -\frac{15}{4} & \frac{19}{4} \\ \frac{5}{2} & \frac{5}{2} & 1 & -2 & \frac{5}{2} \end{pmatrix}$$

```
Eigensystem[A]
```

$$\left\{\{-2, -2, 1, 1, 1\}, \left\{\left\{-\frac{1}{2}, \frac{1}{2}, \frac{1}{2}, \frac{5}{2}, 1\right\}, \{0, 0, 0, 0, 0\}, \{0, 1, 0, 2, 1\}, \{0, 0, 0, 0, 0\}, \{0, 0, 0, 0, 0\}\right\}\right\}$$

We have only two different eigenvalues. The eigenvalue 2 has multiplicity 2 and only a single eigenvector. The eigenvalue 1 has multiplicity 3 and again only a single eigenvector. We will need all of our machinery for repeated eigenvalues. We borrow some of the code from the notebook on repeated eigenvalues.

```
Id := IdentityMatrix[ndim]
```

```
solmake[A_, λ_, m_, v0_, t_] := Module[{ans, i, term},
  ans = Id; Sum[term = MatrixPower[(A - λ * Id), i] * ((t^i) / (i!));
   ans = ans + term, {i, 1, m - 1}]; ans = Exp[λ * t] * (ans.v0); ans]
```

In order to use this code, we must find, for each distinct eigenvalue, the generalized eigenvectors. We start with $\lambda = -2$.

```
ndim = 5;
```

```
λ = -2;
```

```
m = 2;
```

```
M = MatrixPower[(A - λ Id), m]
```

$$\left\{\left\{\frac{21}{4}, 0, 3, \frac{9}{4}, -\frac{9}{2}\right\}, \left\{\frac{29}{4}, 9, \frac{7}{2}, -\frac{21}{4}, \frac{21}{2}\right\}, \left\{\frac{15}{4}, 0, 6, -\frac{9}{4}, \frac{9}{2}\right\}, \left\{\frac{79}{4}, 18, 10, -\frac{33}{4}, \frac{33}{2}\right\}, \left\{11, 9, \frac{19}{2}, -\frac{15}{2}, 15\right\}\right\}$$

```
{v1, v2} = NullSpace[M]
```

$$\{\{2, -2, -2, 0, 1\}, \{-1, 1, 1, 1, 0\}\}$$

```
X1 = solmake[A, λ, m, v1, t]
```

$$\left\{e^{-2t}\left(-\frac{21t}{4}+2(1+2t)\right),\ e^{-2t}\left(\frac{27t}{4}-2\left(1+\frac{11t}{4}\right)\right),\ e^{-2t}\left(\frac{17t}{4}-2\left(1+\frac{3t}{2}\right)\right),\ \frac{25}{4}e^{-2t}t,\ e^{-2t}\left(1+\frac{5t}{2}\right)\right\}$$

```
X2 = solmake[A, λ, m, v2, t]
```

$$\left\{e^{-2t}\left(-1+\frac{t}{2}\right),\ e^{-2t}\left(1-\frac{t}{2}\right),\ e^{-2t}\left(1-\frac{t}{2}\right),\ e^{-2t}\left(1-\frac{5t}{2}\right),\ -e^{-2t}t\right\}$$

This completes the part of the solution basis associated with λ = -2. Now we construct the part of the basis associated with $\lambda = 1$.

```
λ = 1;
```

```
m = 3;
```

```
M = MatrixPower[(A - λ Id), m]
```

$$\left\{\left\{-\frac{27}{4},\frac{27}{4},\frac{81}{4},\frac{27}{4},-\frac{81}{4}\right\},\left\{\frac{27}{4},-\frac{27}{4},-\frac{81}{4},-\frac{27}{4},\frac{81}{4}\right\},\left\{\frac{27}{4},-\frac{27}{4},-\frac{81}{4},-\frac{27}{4},\frac{81}{4}\right\},\left\{\frac{135}{4},\frac{81}{4},-\frac{189}{4},-\frac{135}{4},\frac{189}{4}\right\},\left\{\frac{27}{2},\frac{27}{2},-\frac{27}{2},-\frac{27}{2},\frac{27}{2}\right\}\right\}$$

```
{v3, v4, v5} = NullSpace[M]
```

{{-2, 1, 0, 0, 1}, {1, 0, 0, 1, 0}, {2, -1, 1, 0, 0}}

```
X3 = solmake[A, λ, m, v3, t]
```

$$\left\{e^{t}\left(-\frac{3t}{2}+\frac{9t^2}{4}-2\left(1-t+\frac{9t^2}{8}\right)\right),\ e^{t}\left(1+\frac{5t}{2}-\frac{9t^2}{4}-2\left(\frac{3t}{2}-\frac{7t^2}{8}\right)\right),\ e^{t}\left(\frac{3t}{2}-\frac{9t^2}{4}-2\left(t-\frac{9t^2}{8}\right)\right),\ e^{t}\left(\frac{19t}{2}-\frac{45t^2}{4}-2\left(5t-\frac{41t^2}{8}\right)\right),\ e^{t}\left(1+4t-\frac{9t^2}{2}-2\left(\frac{5t}{2}-2t^2\right)\right)\right\}$$

```
X4 = solmake[A, λ, m, v4, t]
```

$$\left\{e^{t}\left(1-\frac{t}{4}\right),\ e^{t}\left(\frac{t}{4}+\frac{t^2}{4}\right),\ \frac{e^{t}t}{4},\ e^{t}\left(1+\frac{t}{4}+\frac{t^2}{2}\right),\ e^{t}\left(\frac{t}{2}+\frac{t^2}{4}\right)\right\}$$

```
X5 = solmake[A, λ, m, v5, t]
```

$$\left\{e^t\left(\frac{5t}{4}-\frac{9t^2}{4}+2\left(1-t+\frac{9t^2}{8}\right)\right),\ e^t\left(-1-\frac{t}{4}+\frac{5t^2}{2}+2\left(\frac{3t}{2}-\frac{7t^2}{8}\right)\right),\ e^t\left(1-\frac{5t}{4}+\frac{9t^2}{4}+2\left(t-\frac{9t^2}{8}\right)\right),\ e^t\left(-\frac{21t}{4}+\frac{47t^2}{4}+2\left(5t-\frac{41t^2}{8}\right)\right),\ e^t\left(-\frac{3t}{2}+\frac{19t^2}{4}+2\left(\frac{5t}{2}-2t^2\right)\right)\right\}$$

Now we form the general solution, and then find the constants to satisfy the initial condition.

```
Xgen = a1 * X1 + a2 * X2 + a3 * X3 + a4 * X4 + a5 * X5
```

$$\Bigg\{a4\, e^t\left(1-\frac{t}{4}\right)+a2\, e^{-2t}\left(-1+\frac{t}{2}\right)+a1\, e^{-2t}\left(-\frac{21t}{4}+2\,(1+2t)\right)+a3\, e^t\left(-\frac{3t}{2}+\frac{9t^2}{4}-2\left(1-t+\frac{9t^2}{8}\right)\right)+a5\, e^t\left(\frac{5t}{4}-\frac{9t^2}{4}+2\left(1-t+\frac{9t^2}{8}\right)\right),$$

$$a2\, e^{-2t}\left(1-\frac{t}{2}\right)+a4\, e^t\left(\frac{t}{4}+\frac{t^2}{4}\right)+a1\, e^{-2t}\left(\frac{27t}{4}-2\left(1+\frac{11t}{4}\right)\right)+a3\, e^t\left(1+\frac{5t}{2}-\frac{9t^2}{4}-2\left(\frac{3t}{2}-\frac{7t^2}{8}\right)\right)+a5\, e^t\left(-1-\frac{t}{4}+\frac{5t^2}{2}+2\left(\frac{3t}{2}-\frac{7t^2}{8}\right)\right),$$

$$a2\, e^{-2t}\left(1-\frac{t}{2}\right)+\frac{1}{4}\, a4\, e^t\, t+a1\, e^{-2t}\left(\frac{17t}{4}-2\left(1+\frac{3t}{2}\right)\right)+a3\, e^t\left(\frac{3t}{2}-\frac{9t^2}{4}-2\left(t-\frac{9t^2}{8}\right)\right)+a5\, e^t\left(1-\frac{5t}{4}+\frac{9t^2}{4}+2\left(t-\frac{9t^2}{8}\right)\right),$$

$$a2\, e^{-2t}\left(1-\frac{5t}{2}\right)+\frac{25}{4}\, a1\, e^{-2t}\, t+a4\, e^t\left(1+\frac{t}{4}+\frac{t^2}{2}\right)+a3\, e^t\left(\frac{19t}{2}-\frac{45t^2}{4}-2\left(5t-\frac{41t^2}{8}\right)\right)+a5\, e^t\left(-\frac{21t}{4}+\frac{47t^2}{4}+2\left(5t-\frac{41t^2}{8}\right)\right),$$

$$-a2\, e^{-2t}\, t+a1\, e^{-2t}\left(1+\frac{5t}{2}\right)+a4\, e^t\left(\frac{t}{2}+\frac{t^2}{4}\right)+a3\, e^t\left(1+4t-\frac{9t^2}{2}-2\left(\frac{5t}{2}-2t^2\right)\right)+a5\, e^t\left(-\frac{3t}{2}+\frac{19t^2}{4}+2\left(\frac{5t}{2}-2t^2\right)\right)\Bigg\}$$

```
Xinit = {1, 0, 2, -1, 1};
```

```
coeff = Solve[(Xgen /. t → 0) == Xinit, {a1, a2, a3, a4, a5}]
```

$$\{\{a1\to -1,\ a2\to -2,\ a3\to 2,\ a4\to 1,\ a5\to 2\}\}$$

Now we form the solution.

```
Xsol5 = Simplify[Xgen /. Thread[Flatten[coeff]]]
```

$$\left\{\frac{1}{4} e^{-2t}\left(e^{3t}(4-3t)+t\right), \frac{1}{4} e^{-2t} t\left(-1+e^{3t}(19+3t)\right), \frac{1}{4} e^{-2t}\left(-t+e^{3t}(8+3t)\right), \frac{1}{4} e^{-2t}\left(-8-5t+e^{3t}\left(4+35t+6t^2\right)\right), \frac{1}{4} e^{-2t}\left(-2(2+t)+e^{3t}\left(8+22t+3t^2\right)\right)\right\}$$

We check this.

```
Simplify[D[Xsol5, t] - A.Xsol5]
```

{0, 0, 0, 0, 0}

```
Simplify[(Xsol5 /. t → 0) - Xinit]
```

{0, 0, 0, 0, 0}

Everything checks.

We make a plot of the first component for t in the range [0,1].

```
graph1 = Plot[Xsol5[[1]], {t, 0, 2},
  AxesLabel → {"t", "X[[1]]"}, PlotLabel → "Analytical Solution",
  AspectRatio → 0.7, PlotRange → {-5, 1.2}]
```



Now we solve the same problem with DynPac.

```
setstate[{x1, x2, x3, x4, x5}]; setparm[{}];
```

```
slopevec = A.statevec
```

$$\left\{\frac{x2}{4}+\frac{3\,x3}{2}+\frac{3\,x4}{4}-\frac{7\,x5}{4},\ \frac{3\,x1}{2}+\frac{3\,x2}{4}-\frac{x3}{2}-\frac{5\,x4}{4}+\frac{11\,x5}{4},\ x1-\frac{x2}{4}-\frac{x3}{2}-\frac{3\,x4}{4}+\frac{7\,x5}{4},\ 5\,x1+\frac{19\,x2}{4}-\frac{x3}{2}-\frac{15\,x4}{4}+\frac{19\,x5}{4},\ \frac{5\,x1}{2}+\frac{5\,x2}{2}+x3-2\,x4+\frac{5\,x5}{2}\right\}$$

```
t0 = 0.0; h = 0.01; nsteps = 200;
```

```
sol5 = integrate[Xinit, t0, h, nsteps];
```

```
asprat = 0.7; plrange = {{0, 2}, {-5, 1.2}};
```

```
sysname = "Numerical Solution";
```

```
graph2 = timeplot[sol5, 1];
```

```
Show[graph1, graph2]
```



We see that the graphs are essentially identical. Let's compare values of the solutions at $t = 2$.

```
N[Xsol5 /. t → 2]
```

{-3.68537, 92.354, 25.8525, 180.949, 118.188}

```
lastx
```

{-3.68537, 92.354, 25.8525, 180.949, 118.188}

We see perfect agreement to this number of places.